Tree Recursion, Operation Counts, Tail Recursion, and Variable Scope

Lecturer:

Professor Albert R. Meyer

September 11, 1997

More Recursion by Wishful Thinking

Fibonacci numbers: 0 1 1 2 3 5 8 13 21 . . .

Each number is the sum of the two numbers preceding it Useful facts:

fibn = fibn�_1 + fibn�_2 fib_0 = 0 fib_1 = 1

Recursive Fibonacci Procedure

(define (fibonacci n)
  (if (< n 2)
      n
      (+ (fibonacci (- n 1))
         (fibonacci (- n 2)))))

How many primitive operations to evaluate Fib(n)?

Operation Count Mantra for Combinations

(⟨operation⟩ ⟨arg_1⟩ . . . )

The number of operations needed to find the value of a combination is the sum of

• the number of operations needed to find the values of ⟨operator⟩ and ⟨arg⟩’s, and

• the number of ops for the application of the value of ⟨operator⟩ to the values of the ⟨args⟩’s.

Operation Count Mantra for Compound Proc Application

The number of ops for the application of a compound procedure named

proc applied to values M, N in an environment where proc is defined by

                   (define (proc x y) ⟨body⟩)

is the number of ops for ⟨body⟩ in the environment with x given the value M, and y given the value N.

How many primitive operations to evaluate Fib(n)? #ops(0) = #ops(1) = 1 just the “<” #ops(n) = #ops(n − 1) + #ops(n − 2) + 4 (the 4 ops are one “<”, one “+”, and two “-”) But clearly #ops(n − 1) ≥ #ops(n − 2) so #ops(n) > 2 · #ops(n − 2) > 2 · (2 · #ops(n − 4)) > 2n=^2 = ( √ 2)n > (1.4)n EXPONENTIAL growth!

               fibonacci Trace

(trace-entry fibonacci)
(fibonacci 4)
[fibonacci Args: 4]
[fibonacci Args: 2]
[fibonacci Args: 0]
[fibonacci Args: 1]
[fibonacci Args: 3]
[fibonacci Args: 1]
[fibonacci Args: 2]
[fibonacci Args: 0]
[fibonacci Args: 1]
; Value: 3

Smarter, faster Fib is based on the idea of keeping the values of Fib(k − 1) when doing Fib(k). Say a is Fib(k − 1) and b is Fib(k − 2); then keep updating

a := a + b b := a } simultaneously

Faster Fibonacci

(define (fast-fib n)
  (helper 1 0 n))

(define (helper a b n)
  (if (zero? n)
      b
      (helper (+ a b) a (- n 1))))

Now,

#ops(n) = 3 + #ops(n − 1) = 3 + (3 + #ops(n − 2)) . . . = 3n + 1

LINEAR Growth!

                fast-fib trace

(trace-entry helper)
(fast-fib 5)
[helper Args: 1 0 5]
[helper Args: 1 1 4]
[helper Args: 2 1 3]
[helper Args: 3 2 2]
[helper Args: 5 3 1]
[helper Args: 8 5 0]
; Value: 8

Another example

Smart exponentiation: bn Useful facts: b^0 = 1 b^2n = (bn)^2 bn^{+1} = b ∗ bn

Fast Exponentiation

(define (fast-expt b n)
  (if (zero? n)
      1
      (if (even? n)
          (square (fast-expt b (/ n 2)))
          (* b (fast-expt b (- n 1))))))

fast-expt operations

If n is even,

#ops(n) = 4 + #ops(n/2)

If n is odd,

#ops(n) = 4 + #ops(n − 1) = 4 + ( 4 + #ops ( n − 1 2 ))

So in either case,

#ops(n) ≤ 8 + #ops(⌊n/2⌋) ≤ 8 + #ops (8 + #ops (⌊⌊n/2⌋ /2⌋)) ≤ 8 + (8 + #ops(⌊n/4⌋)) ≤ 8 · log_2 n + 1

LOGARITHMIC growth!

               fast-expt trace

(trace-exit fast-expt)
(fast-expt 10 30)
[1 fast-expt Args: 10 0]
[10 fast-expt Args: 10 1]
[100 fast-expt Args: 10 2]
[1000 fast-expt Args: 10 3]
[1000000 fast-expt Args: 10 6]
[10000000 fast-expt Args: 10 7]
[100000000000000 fast-expt Args: 14]
[1000000000000000 fast-expt Args: 15]
[1000000000000000000000000000000 fast-expt Args: 30]
; Value: 1000000000000000000000000000000

Variable Scope (dropping parameter b)

(define (fast-expt b n)
  (define (helper m)
    (if (zero? m)
        1
        (if (even? m)
            (square (helper (/ m 2)))
            (* b (helper (- m 1))))))
  (helper n))

Also, helper is hidden outside the definition of fast-expt.

Internal Defines: Augmented Value Mantra for Application of Compound Procedures

“The value of a compound procedure named proc applied to a value, n, in an environment where proc is defined by

(define (proc x)
  (define (proc-helper y) <body>)
  <Sch-expr>)

is the value of ⟨Sch-expr⟩ in the environment with x given the value n, and with a proc-helper procedure defined by the internal definition.”

Importance of Scheme’s Scoping Rules

ffl Procedure parameters and internal definitions are invisible outside the procedure. Redefining them outside the procedure will not affect the procedure’s behaviour in any way.

ffl Renaming an internal name into any “fresh” name—a name which is not already an internal name—will not affect the computational behavior of anything at all.

Tail Recursive Procedures

Easy to simulate step-by-step, remembering only two things:

• a single position in the code being evaluated, and • the values of the variables mentioned in the code.

Summary: Scheme Syntax so far

• Keywords: if, define • Boolean and Number Constants • Variables: names which are not Keywords or Constants • Scheme Expressions:

{ constants, { variables, { if expressions, or { combinations.

• if expression:

(if ⟨predicate⟩ ⟨consequent⟩ ⟨alternative⟩)

where ⟨predicate⟩, . . . , are Scheme expressions. • . . .

Complete Scheme Syntax So Far, BNF notation:

⟨keyword⟩ ::= if, define, . . . ⟨Boolean constants⟩ ::= #t, #f ⟨Number constant⟩ ::= 2, −3.7, 792e14, . . .

⟨Sch-expr⟩ ::= ⟨constant⟩ | ⟨variable⟩ | ⟨if-exp⟩ | ⟨combination⟩ ⟨if-expr⟩ ::= (if ⟨Sch-expr⟩ ⟨Sch-expr⟩ ⟨Sch-expr⟩) ⟨combination⟩ ::= (⟨expression_1⟩ ⟨expression_2⟩ . . . ) ⟨define⟩ ::= (define (⟨variable⟩ ⟨formals⟩) ⟨body⟩) | (define ⟨variable⟩ ⟨Sch-expr⟩) ⟨formals⟩ ::= ⟨variable_1⟩ ⟨variable_2⟩ . . .

⟨body⟩ ::= ⟨Sch-expr⟩ | ⟨defines⟩ ⟨Sch-expr⟩ ⟨defines⟩ ::= ⟨define⟩ | ⟨define⟩ ⟨defines⟩

(No repeated define’s of same variable.)

Syntactic Definition of Tail-Recursion

⟨simple-expr⟩ ::= ⟨constant⟩ | ⟨variable⟩ | (⟨primitive-operator⟩⟨simple-expr_1⟩⟨simple-expr_2⟩ . . . )

⟨t-r-expr⟩ ::= ⟨simple-expr⟩ | (⟨variable⟩ ⟨simple-expr_1⟩ ⟨simple-expr_2⟩ . . . ) | (if ⟨simple-expr⟩ ⟨t-r-expr⟩ ⟨t-r-expr⟩) ⟨t-r-def⟩ ::= (define (⟨variable⟩ ⟨formals⟩) ⟨t-r-body⟩) |

(define ⟨variable⟩ ⟨t-r-expr⟩)

⟨t-r-body⟩ ::= ⟨t-r-defines⟩⟨t-r-expr⟩ ⟨t-r-defs⟩ ::= ⟨t-r-def⟩ | ⟨t-r-def⟩⟨t-r-defs⟩

Summary: Value and Op Count Mantras

• Mantras are rules for deducing the value of an expression and the number of operations required by Scheme to calculate that value.

• Mantras do not model step-by-step behavior of Scheme evalulation, and do not model computational space requirements. (Next lecture on the Substitution Model will provide these.)

• Mantras are correct: values and op counts from mantras do agree with “real” Scheme.

Tail Recursive Procedures

• Very restricted procedure definition format.

• Have a simple step-by-step computational model: move a “finger” along the text of definition.

• Computational Space for a procedure application is bounded by the size of the text of the definition. The space used is a fixed

constant independent of the value to which the procedure is applied!

Scoping

• Protects privacy of names (creates abstraction) • Can reduce number of parameters